О св-вах освещения и материалах.
Источник света по умолчанию располагается в пространстве в точке с координатами (0, 0, 1). Напомню если режим GL_COLOR_MATERIAL не включен, то текущие цветовые установки не влияют на цвет примитивов. Позицию источника света задаем так : объявим переменную LightPos : Array [0..3] of GLfloat; инициализируем в нее нужные нам координаты , и теперь задаем позицию glLightfv(GL_LIGHT0, GL_POSITION, @LightPos); аргументы : имя источника , константа символизирующая нужный параметр , ссылка на массив с координатами. Символьные константы GL_AMBIENT, GL_DIFFUSE и GL_SPECULAR, (рассеянный , диффузный , зеркальный цвет) указанные в качестве второго аргумента команды glLight, позволяют задавать требуемые свойства окружающей среды. Для этого тоже нужен массив , как и выше. Вес цвета в диффузной составляющей задает, насколько сильно этот цвет отражается поверхностью при ее освещении. И наоборот, вес цвета в фоновой составляющей задает, насколько сильно этот цвет поглощается поверхностью. По умолчанию поверхность ничего не поглощает и все отражает. Объясняю на примере : задайте фоновое отражение каким-либо чистым цветом, например красным - после этого наименее освещенные части поверхности примитива окрашиваются этим цветом. Если значения цветового фильтра, RGB, тоже установить красными, вся сфера окрасится равномерно. Установите значения диффузного отражения в тот же чистый цвет. Теперь в этот цвет окрашиваются участки поверхности сферы, наиболее сильно освещенные источником света. Слабо освещенные участки окрашены в оттенки серого. Если RGB установить в этот же цвет, насыщенность им увеличивается по всей поверхности. Если обе характеристики задать равными, поверхность равномерно окрашивается ровным светлым оттенком. Короче попробуйте сами. Свойства материала задаются с помощью команды glMaterial . Первый аргумент – стороны треугольников для которых задаются характеристики. Характеристики, определяющие оптические свойства материала, и соответствующие им символьные константы являются следующими: рассеянный цвет (GL_AMBIENT), диффузный цвет (GL_DIFFUSE), зеркальный цвет (GL_SPECULAR), излучаемый цвет (GL_EMISSION), степень зеркального отражения (GL_ SHININESS). Значением последнего параметра может быть число из интервала [0,128] (размер блика), oстальные параметры представляют собой массивы четырех вещественных чисел. Зеркальный цвет задает цветовую гамму бликов материала, степень зеркального отражения определяет, насколько близка поверхность к идеальному зеркалу. При заданных характеристиках материала GL_COLOR_MATERIAL можно не включать. Для того чтобы включить освещенность для внутренней стороны многоугольников, вызывается команда glLightModel, у которой вторым аргументом задается символическая константа GL_LIGHT_MODEL_TWO_SIDE, а третьим аргументом - ноль или единица. Кстати вас не озадачило , зачем для определения координат источника света массив из 4-х чисел ? Упрощенно этот параметр можно назвать как дальность действия источника света. При значении 0 на степень освещения не влияет расстояние между источником и примитивом. Короче это степень затухания света. Команду glMaterial можно заменить так: glColorMaterial(GL_FRONT, GL_DIFFUSE); glEnable(GL_COLOR_MATERIAL); и теперь цвет примитива ( команда glColor ) устанавливает не общий цвет примитива , а только диффузную составляющую. С точки зрения оптимизации эта команда предпочтительнее команды glMaterial. Если хотите разобраться с материалами , то можете поэкспериментировать на примере из прошлой главы. 


Немного отвлечемся : По умолчанию обход вершин против часовой стрелки задает лицевую сторону многоугольника. Команда glFrontFace (GL_CW); меняет правило обхода на противоположный. 


Прозрачность , смешение цветов , буфер трафарета (stencil), буфер накопления
Так называемый альфа-компонент, задаваемый четвертым аргументом в командах, связанных с цветом фона и примитивов, используется для получения смеси цветов.
Режим смешения включается командой glEnable с аргументом GL_BLEND Помимо этой команды необходимо задать пиксельную арифметику с помощью glBlendFunc Аргументы этой команды определяют правила, по которым происходит смешение поступающих значений RGBA с содержимым буфера кадра. glBlendFunc (GL_SRC_ALPHA, GL_ONE_MINUS_SRC_ALPHA) ; - наиболее часто употребляемая функция. Естественно четвертый компонент в массивах , задающих цвета – степень прозрачности (0-не прозрачен , 1-полностью прозрачен).

В динамике у некоторых объектов может появляться ненужный узор , от которого надо избавиться. Он связан с наложением друг на друга передней и задней полупрозрачной поверхности объекта. Можно не воспроизводить задние стороны треугольников , но если все-таки нам это надо , то мы воспроизводим сначала задние стороны , потом передние. glCullFace(GL_FRONT); не путайте – так мы задаем какую сторону НЕ выводим . В некоторых случаях возможно придется отключать буфер глубины. Если на сцене несколько полупрозрачных объектов ? Тогда надо каким-либо способом определять какой предмет за каким , и воспроизводить сначала наиболее удаленный объект.

Многие специальные эффекты, самым простым из которых является вырезка объекта, основаны на использовании буфера трафарета (шаблона). Напомню, что одним из полей в формате пиксела является размер буфера трафарета, который надо задавать в соответствии с характеристиками вашей графической платы

Выполнение теста трафарета разрешается командой glEnable с аргументом GL_STENCIL_TEST. Команда glstencilFunc отвечает за сравнение, а команда glstencilop позволяет определить действия, базирующиеся на результате проверки трафарета. Начинаем работу с буфером так: 

glClearStencil(0); // значение заполнения буфера трафарета при очистке
glStencilMask(l); // число битов, задающее маску
glEnable(GL_STENCIL_TEST); // включаем тест буфера трафарета

Первые две команды здесь даны с значениями по умолчанию , их можно удалить. В большинстве случаев используют значение по умолчанию. Аргументы этих команд имеют смысл битовых масок, поэтому корректнее задавать их шестнадцатеричными. Первая команда задает фоновое значение, которым будет заполнен буфер при выполнении команды glclear с аргументом GL_STENCIL_BUFFER_BIT , вторая команда разрешает или запрещает перезапись битов в плоскости трафарета. Основу примера напишите сами. Мы разберем код перерисовки кадра: //очищаются буфер цвета и буфер трафарета


glClear (GL_COLOR_BUFFER_BIT or GL_STENCIL_BUFFER_BIT);

// треугольник

// тест всегда завершается положительно


glstencilFunc(GL_ALWAYS, 1, 1);
// значение буфера устанавливается в 1 для всех точек треугольника


glStencilOp(GL_KEEP, GL_KEEP, GL_REPLACE);
glColor3ub(200, 0, 0); // цвет - красный

glBegin(GL_POLYGON); 


glVertex3i(-4, -4, 0); glVertex3i( 4, -4, 0); glVertex3i( 0, 4, 0) ; 
glEnd;
// зеленый квадрат

// только для точек, где в буфере записана единица


glStencilFunc(GL_EQUAL, 1, 1),
// для точек, не подпадающих в тест, значение буфера установить в максимальное значение, если тест завершился удачно, но в буфере глубины меньшее значение, сохранить текущее значение буфера трафарета, если в буфере глубины большее значение, задать значение нулевым


glStencilOp(GL_INCR, GL_KEEP, GL_DECR);
glColor3ub(0, 200, 0); glBegin(GL_POLYGON);
glVertex3i(3, 3, 0}; glVertex3i(-3, 3, 0); glVertex3i(-3, -3, 0); glVertex3i(3, -3, 0); glEnd;
// синий квадрат

// только для точек, где в буфере записана единица

glstencilFunc(GL_EQUAL, 1, 1);

// для всех точек сохранить текущее значение в буфере трафарета


glStencilOp (GL_KEEP, GL_KEEP, GL_KEEP) ;
glColor3ub(0, 0, 200); 
glBegin(GL_POLYGON);
glVertex3i(3, 3, 0) ; glVertex3i(-3, 3, 0);  glVertex3i(-3, -3, 0);  glVertex3i(3, -3, 0); glEnd;

В данном случае во всех точках треугольника значение буфера трафарета задается равным единице. Для зеленого квадрата тестирование завершается успешно только в точках, где уже записана единица, т е там, где был нарисован треугольник. Во всех точках, выпадающих за границу треугольника, значение буфера трафарета будет задано в максимальное значение, но эти участки не будут нарисованы. Синий квадрат рисуется также с тестированием в точках, где записана единица, т.е там, где рисовался треугольник или предыдущий квадрат Аргументы команды glstencilOp перед воспроизведением синего квадрата уже никак не влияют на характер изображения. Во многих случаях надо отключать буфер глубины. Замечу, что если необходимо на время отключать этот буфер глубины можно воспользоваться командой glDepthMask c аргументом False. Отключать цвета можно так glColorMask(False, False, False, False); и включать понятно как. Короче если раскинуть мозгами , можно понять , что с помощью буфера трафарета можно делать вырезки , и другие интересные вещи , основанные на том , что отрисовываться может только часть пикселов экрана , которым присвоено нужное нам значение в буфере трафарета. Применительно к играм , буфер трафарета используют для визуализации динамических теней , которые так и называются – stencil-shadow. Огорчу вас – здесь про них вы ничего не узнаете , поскольку мне самому надо с ними разбираться (сам алгоритм не сложный , но там очень много тонких нюансов , может в будущем напишу). Вообще буфер трафарета заслуживает отдельной статьи , отсюда вы узнаете , только что это такое.

Буфер накопления аналогичен буферу цвета, но при выводе в него образы накапливаются и накладываются друг на друга. Используется буфер для получения эффектов нерезкости и сглаживания. Поставим перед собой задачу : нарисовать 2 частично перекрывающихся прямоугольника. В области пересечения происходит смешивание цветов. 


glClear (GL_COLOR_BUFFER_BIT) ;

glCallList(thing1); // Какой-либо прямоугольник 

буфер кадра загружается в буфер накопления с коэффициентом 0.5


glAcсum(GL_LOAD, 0.5);
glClear(GL_COLOR_BUFFER_BIT); // экран очищается

glCallList(thing2); // зеленый прямоугольник

// наложение старого содержимого буфера накопления

//с содержимым буфера кадра


glAcсum(GL_ACСUM, 0.5);
// содержимое буфера накопления выводится в буфер кадра


glAcсum(GL_RETURN, 1.0);
Здесь требуются некоторые пояснения. Если вторым аргументом команды glAcсum задается GL_LOAD, старое содержимое буфера аккумуляции затирается, подменяется содержимым буфера кадра. Первые три действия примера можно осуществлять и так, т. е. явным образом очищая буфер накопления. 

// очищаются буфер кадра и буфер накопления


glClear(GL_COLOR_BUFFER_BIT or GL_ACСUM_BUFFER_BIT);

glCallList(thingl); // красный прямоугольник

// содержимое буфера кадра смешивается с содержимым буфера накопления

// (пустым)


glAcсum(GL_ACСUM, 0.5);
Вторым параметром команды glAcсum можно манипулировать для управления яркостью изображения, в данном примере полигоны помещаются в буфер накопления с коэффициентом 0 5, т е их яркость уменьшается вполовину. При итоговом вызове этой команды яркость можно увеличить, коэффициент может принимать значения больше единицы. Чтобы смазать изображение, в буфере накопления одна и та же сцена рисуется несколько раз, каждый раз с немного измененной точкой зрения. Например, чтобы изображение двоилось, надо нарисовать сцену два раза под двумя точками зрения. Чем больше мы возьмем таких точек зрения, тем более ровным окажется размытость, но такие эффекты, конечно, затормаживают воспроизведение. Можно пожертвовать качеством изображения, но повысить скорость взять небольшое количество точек зрения, но увеличить расстояние между ними в пространстве. Чтобы не было искажений цвета , коэффициент заполнения в итоге должен получаться равным 1. Можно добиться эффекта фокуса. Для этого надо не только переносить точку зрения при заполнении буфера новым кадром , но и слегка менять параметры перспективы.

Туман , дымка.


Туман является самым простым в использовании спецэффектом, предназначенным для передачи глубины пространства. Он позволяет имитировать атмосферные эффекты дымки и собственно тумана. Не стоит ожидать хорошего результата от этого тумана. 

Естественно нам понадобиться массив хранящий значение цвета.


glEnable(GL_FOG); // включаем режим тумана
glFogi(GL_FOG_MODE, GL_EXP); // задаем закон смешения тумана
glFogfv(GL_FOG_COLOR, fogColor); // цвет дымки
glFogf(GL_FOG_DENSITY, 0.35); // плотность тумана
glHint(GL_FOG_HINT, GL_DONT_CARE);// предпочтений к работе тумана нет

Помимо константы GL_EXP могут использоваться еще GL_LINEAR и GL_EXP2. Также есть команды glFogf(GL_FOG_START, 1.0); ближняя плоскость затухания, glFogf(GL_FOG_END, 5.0); дальняя плоскость для линейного затухания.


Поверхности произвольной формы , кривые Безье , NURBS , сплайны , патчи .
Понятное дело , достаточно просто строит поверхности из треугольников. Можно организовать систему считывания вершин треугольников из текстовых файлов и цикл отрисовки связанных треугольников. Здесь нужна умная голова , новых знаний не потребуется. Экспорт из других популярных форматов мы не рассматриваем. Но есть другие способы создания произвольных поверхностей. Подход, предлагаемый библиотекой OpenGL для изображения криволинейных поверхностей, традиционен для компьютерной графики: задаются координаты небольшого числа опорных точек, определяющих вид искомой поверхности. В зависимости от способа расчета опорные точки могут лежать на получаемой поверхности, а могут и не располагаться на ней.
Сглаживающие поверхности называются сплайнами. Есть много способов построения сплайнов, из наиболее распространенных нас будут интересовать только два: кривые Безье (Bezier curves, в некоторых книгах называются "сплайны Безье") и В-сплайны (base-splines, базовые сплайны). Операционная система располагает функциями GDI, позволяющими строить кривые Безье. Вершины задаем в массиве четырех величин типа Tpoint, этот тип в модуле windows. pas: 

Const

Points: Array [0.. 3] of TPoint =

((x: 5; y: 5), (x: 20; y: 70), (x: 80; y: 15), (x: 100; y: 90));

Собственно рисование кривой состоит в вызове функции GDI polyBezier, первый аргумент которой - ссылка на контекст устройства, затем указывается массив опорных точек, последний аргумент - количество используемых точек: PolyBezier (dc, Points, 4); Получим двухмерную кривую Безье. Теперь нарисуем кривую Безье средствами OGL. Объявим массив контрольных точек const ctrlpoints : Array [0..3, 0..2] of GLfloat = (( -4.0, -4.0, 0.0), (-2.0, 4.0, 0.0), (2.0, -4.0, 0.0), (4.0, 4.0, 0.0)); В обработчике создания формы задаются параметры так называемого одномерного вычислителя, и включается этот самый вычислитель:


glMap1f (GL_MAP1_VERTEX_3, 0. 0, 1. 0, 3, 4, @ctrlpoints); 
glEnable (GL_MAPl_VERTEX_3);
Первый параметр команды glMap1 - символическая константа, значение GL_MAP1_VERTEX_3 соответствует случаю, когда каждая контрольная точка представляет собой набор трех вещественных чисел одинарной точности, т. e. координаты точки. Значения второго и третьего аргументов команды определяют конечные точки интервала предварительного образа рассчитываемой кривой. Величины ноль и один для них являются обычно используемыми, подробнее мы рассмотрим эти аргументы чуть ниже. Четвертый параметр команды, "большой шаг", задает, сколько чисел содержится в считываемой порции данных. Как говорится в документации, контрольные точки могут содержаться в произвольных структурах данных, лишь бы их значения располагались в памяти друг за другом. Последние два параметра команды - число опорных точек и указатель на массив опорных точек.
Для построения кривой вместо команды, задающей вершину, вызывается команда glEvalcoord, возвращающая координаты рассчитанной кривой:
glBegin(GL__LINE_STRIP); 
For i: = 0 to 30 do
glEvalCoord1f (i / 30. 0); 
glEnd;
Аргумент команды - значение координаты u. В данном примере соединяются отрезками тридцать точек, равномерно расположенных на кривой Теперь выясним правила отображения интервала. Если третий параметр команды glMap1f задать равным двум, то на экране получим половину первоначальной кривой. Для получения полной кривой надо при воспроизведении взять интервал в два раза больший : для получения полной кривой надо при воспроизведении взять интервал в два раза больший: glBegin(GL_LINE_STRIP); For i: = 0 to 60 do glEvalCoord1f(i / 30. 0); glEnd; Если же этот параметр задать равным 0. 5, то при отображении интервала с u в пределах от нуля до единицы получаемая кривая не будет останавливаться на последней опорной точке, а экстраполироваться дальше. Короче надо самим попрактиковаться.

Теперь давайте получим поверхность Безье , а не линию.


Зададим 16 точек (не запутайтесь в скобках) : ctrlpoints : Array [0..3, 0..3, 0..2] of GLFloat = ( ( (-1.5, -1.5, 4.0),(-0.5, -1.5, 2.0),(0.5, -1.5, -1.0),(1.5, -1.5, 2.0)),
  ( (-1.5, -0.5, 1.0),(-0.5, -0.5, 3.0),(0.5, -0.5, 0.0),(1.5, -0.5, -1.0)),
  ( (-1.5, 0.5, 4.0),(-0.5, 0.5, 0.0),(0.5, 0.5, 3.0),(1.5, 0.5, 4.0)),
  ( (-1.5, 1.5, -2.0),(-0.5, 1.5, -2.0),(0.5, 1.5, 0.0),(1.5, 1.5, -1.0)) );
Работа программы начинается с установки параметров вычислителя:
glMap2f (GL_MAP2_VERTEX_3, 0, 1, 3, 4, 0, 1, 12, 4, @ctrlpoints);
glEnable (GL_MAP2_VERTEX_3);
glMapGrid2f (20, 0. 0, 1. 0, 20, 0. 0, 1. 0); 
У команды glMap2f аргументов больше, чем у glMap1f. Первый аргумент - константа, определяющая тип рассчитываемых величин, в данном случае это координаты точек поверхности. Последний аргумент - указатель на массив контрольных точек поверхности.
Второй и третий параметры задают преобразования по координате u поверхности. Четвертый аргумент, как и в предыдущем случае, задает, сколько вещественных чисел содержится в порции данных - здесь мы сообщаем, что каждая точка задана тремя координатами. Пятый аргумент - количество точек в строке структуры, хранящей данные.
Следующие четыре аргумента имеют смысл, аналогичный предыдущим четырем, но задают параметры для второй координаты поверхности, координаты v. Значение восьмого аргумента стало равно двенадцати путем перемножения количества чисел, задающих координаты одной вершины (3), на количество точек в строке массива (4). После задания параметров вычислителя он включается вызовом команды glEnable, после чего вызывается одна из пары команд, позволяющих построить поверхность - команда glMapGrid2f, рассчитывающая двумерную сетку. Первый и четвертый аргументы этой команды определяют количество разбиений по каждой из двух координат поверхности, остальные параметры имеют отношение к отображению интервалов. Собственно изображение поверхности, двумерной сетки, осуществляется вызовом команды glEvalMesh2 (GL_FILL, 0, 20, 0, 20);
Первый аргумент команды задает режим воспроизведения (можно GL_LINE). Следующие две пары чисел задают количество подинтервалов разбиения по каждой координате поверхности. В примере мы берем по 20 разбиений, столько же, сколько задано в команде glMapGrid2f, чтобы не выходить за пределы интервалов, но это не обязательно, можно брать и больше. Замечу, что режим воспроизведения можно варьировать также с помощью команды glPolygonMode. Также желательно использовать режим автоматического расчета нормалей к поверхности: glEnable (GL_AUTO_NORMAL); Без этого режима поверхность выглядит невыразительно, но его использование допускается только в случае поверхностей, рассчитываемых вычислителем. 


Примечание: Командой glIsEnabled можно проверять включен ли какой либо режим или нет. Аргумент команды – названия режимов ( аналогично glEnable ). Результат работы команды – величина типа GLboolean. 
Один из классов В-сплайнов, рациональные В-сплайны, задаваемые на неравномерной сетке (Non-Uniform Rational B-Spline, NURBS), является стандартным для компьютерной графики способом определения параметрических кривых и поверхностей. Библиотека glu предоставляет набор команд, позволяющих использовать этот класс поверхностей.

Я надеюсь вы написали приложение строящее двухмерную кривую Безье. А теперь построим NURBS по тем же контрольным точкам ( не ждите чуда – вид кривой будет почти тот же). Для работы с NURBS-поверхностями в библиотеке glu имеются переменные специального типа, используемые для идентификации объектов:


theNurb: gluNurbsObj;
При создании окна объект, как обычно, создается:


theNurb: = gluNewNurbsRenderer;
А в конце работы приложения память, занимаемая объектом, высвобождается:

gluDeleteNurbsRenderer (theNurb);

Прим.: в отличие от quadric-объектов , NURBS-объекты после удаления действительно недоступны.
построение кривой осуществляется следующей командой:
gluNurbsCurve (theNurb, 8, @curveKnots, 3, @ctrlpoints, 4, GL_MAPl_VERTEX_3 );
Первый аргумент - имя NURBS-объекта, вторым аргументом задается количество параметрических узлов кривой, третий аргумент - указатель на массив, хранящий значения этих узлов. Следующий параметр - смещение, т.e. сколько вещественных чисел содержится в порции данных, далее следует указатель на массив опорных точек. Последний аргумент равен порядку (степени) кривой плюс единица. Количество узлов должно равняться количеству опорных точек плюс порядок кривой. Количество опорных точек достаточно взять на единицу больше степени кривой: квадратичная кривая однозначно определяется тремя точками, кубическая - четырьмя и т. д. Так что для построения квадратичной кривой необходимо задавать шесть узлов:

gluNurbsCurve (theNurb, 6, @curveKnots, 3, @ctrlpoints, 3, GL_MAP1_VERTEX_3);


Вот пример массива хранящего значение параметрических узлов:
curveKnots : Array [0..7] of GLfloat = (0.0, 0.0, 0.0, 0.0, 1.0, 1.0, 1.0, 1.0);

Значения параметрических узлов в массиве или другой структуре, хранящей данные, должны быть упорядочены по неубыванию, т. e. каждое значение не может быть меньше предыдущего. Обычно значения первой половины узлов берутся равными нулю, значения второй половины задаются единичными, что соответствует неискаженной кривой, строящейся на всем интервале от первой до последней опорной точки. Короче здесь надо поэкспериментировать самим. Для манипулирования свойствами таких объектов предусмотрена специальная команда библиотеки, например для задания гладкости поверхности и режима воспроизведения (гладкость кривой или поверхности задается допуском дискретизации: чем меньше это число, тем поверхность получается более гладкой): gluNurbsProperty (theNurb, GLU_SAMPLING_TOLERANCE, 25. 0);
В отличие от других объектов, NURBS-поверхности рассчитываются каждый раз заново при каждом построении.


Построим NURBS-поверхность.

Первым делом замечу, что здесь нужен новый для нас режим:


glEnable (GL_NORMALIZE); Сделано это из-за того, что поверхность при построении масштабируется, и чтобы автоматически рассчитанные нормали "не уплыли", и используется этот режим. Две команды ниже позволяют менять режим воспроизведения.
gluNurbsProperty(theNurb, GLU_DISPLAY_MODE, GLU_FILL)
gluNurbsProperty(theNurb, GLU_DISPLAY_MODE, GLU_OUTLINE_POLYGON);
Нижний режим – просто сетка , верхний – полностью.
Команда собственно построения заключим в специальные командные скобки:
gluBeginSurface (theNurb); 
gluNurbsSurface (theNurb,8, @knots,8, @knots,4 * 3,3,@ctrlpoints,4, 4, GL__MAP2_VERTEX_3); 
gluEndSurface (theNurb);
В данном примере эти скобки не обязательны, поскольку они обрамляют единственную команду. Шестой и седьмой параметры задают "большой шаг" по каждой координате, ассоциированной с поверхностью, т.e. сколько вещественных чисел содержится в строке структуры данных и сколько вещественных чисел задают отдельную точку. Восьмой параметр - адрес массива контрольных точек, а последним параметром задается символическая константа, определяющая тип возвращаемых значений; в данном случае ими являются трехмерные координаты вершин. Поверхность Безье по виду почти идентична NURBS-поверхности. Отличает NURBS-поверхности то, что параметризируемы. Так, предпоследние два параметра задают степень (порядок) поверхности по координатам u и v. Задаваемое число, как сказано в документации, должно быть на единицу больше требуемой степени. Для поверхности, кубической по этим координатам, число должно равняться 4, как в нашем примере. Порядок нельзя задавать совершенно произвольным, ниже мы разберем имеющиеся ограничения. Второй параметр команды - количество узлов в параметризации по направлению, третьим параметром задается адрес массива, хранящего значения узлов. Третий и четвертый параметры команды имеют аналогичный смысл, но для второго направления.
Массивы узлов должны заполняться значениями, упорядоченными по неубыванию. Как сказано в файле справки, при заданных uknot_count и vknot_count количествах узлов, uorder и vorder порядках количество опорных точек должно ровняться (uknot_count - uorder) x (vknot_count - vorder). Так что при изменении порядка по координатам необходимо подбирать и все остальные параметры поверхности. Если вы хотите подробнее узнать о параметризации, и ее параметрах, то обратитесь к литературе с подробным изложением теории NURBS-поверхностей. 

Чаще всего используются "кубические" NURBS-поверхности. Библиотека glu предоставляет также набор команд для вырезки кусков NURBS-поверхностей. 

Внутри операторных скобок построения NURBS-поверхности вслед за командой gluNurbsSurface задаем область вырезки:


gluBeginTrim (theNurb) ;
gluPwlCurve (theNurb, 21, @edgePt, 2, GLU_MAP1_TRIM_2) ; 
gluEndTrim (theNurb);
При задании области вырезки используются опять-таки специальные командные скобки, собственно область вырезки задается командой gluPwlCurve. Команда задает замкнутую кусочно-линейную кривую, часть NURBS-поверхности, не вошедшая внутрь этой области, не воспроизводится. Второй аргумент - количество точек границы, третий - адрес массива этих точек, четвертым параметром является символьная константа, задающая тип вырезки. 

NURBS очень интересная тема с которой вам следует попробовать разобраться по лучше самим , а мы двинемся дальше.


Одним из главных достоинств сплайнов является то, что поверхность задается небольшим количеством опорных точек - весьма экономный подход. Возможно, к достоинствам можно отнести и то, что нет необходимости самостоятельно рассчитывать нормали к поверхности , в отличие от поверхностей из треугольников. Создавать кусочки поверхностей с помощью сплайнов - дело несложное, если попытаться подобрать опорные точки для того, чтобы нарисовать одним сплайном поверхности подобные чайнику, то это окажется занятием крайне трудоемким. Если же поверхность имеет негладкие участки, как, например, область соприкосновения носика чайника с его корпусом, то традиционными сплайнами это нарисовать окажется вообще невозможно. Имеется стандартный способ рисования поверхностей, сочетающий в себе достоинства использования сплайнов и лишенный его недостатков: поверхность разбивается на отдельные гладкие участки, для каждого из которых строится отдельный сплайн. Такие кусочки называются патчами (patch, заплатка). Сейчас мы разберем пример , в котором с помощью патчей строится модель розы. Стрелками вы сможете вращать модель. Пример узко направленный , позволяет загружать только данную модель из данного текстового файла. Не более.
///////////////////////////////////////////////////////////////////////////////////////////////
unit frmMain;
interface
uses Windows, Messages, Classes, Graphics, Forms, ExtCtrls, Menus, Controls, SysUtils, Dialogs, OpenGL;
 type TVector = record // тип хранящий данные о патчах(конкретно – 1-ой опорной точки)
         x, y, z : GLfloat;  
         end;
  TPatch = Array [0..24] of TVector;  //каждый патч состоит из 25 точек 
type  TfrmGL = class(TForm)
    procedure FormCreate(Sender: TObject);  procedure FormResize(Sender: TObject);
    procedure FormDestroy(Sender: TObject);
    procedure FormKeyDown(Sender: TObject; var Key: Word;Shift: TShiftState);
  private
    DC: HDC;  hrc: HGLRC;
    AngX, AngY, AngZ : GLfloat;
    procedure Init;
    procedure init_surface;
    procedure SetDCPixelFormat;
  protected procedure WMPaint(var Msg: TWMPaint); message WM_PAINT;
  end;
const  ROZA = 1;  // имя для списка
var     frmGL: TfrmGL;
implementation
{$R *.DFM}
procedure TfrmGL.Init; //надеюсь данная процедура понятна , иначе см. прошлые статьи.
const
  ambient : Array [0..3] of GLFloat = (0.2, 0.2, 0.2, 1.0);
  position1 : Array [0..3] of GLFloat = (0.0, 2.0, -2.0, 0.0);
  position2 : Array [0..3] of GLFloat = (0.0, 2.0, 5.0, 0.0);
  mat_diffuse : Array [0..3] of GLFloat = (1.0, 0.0, 0.0, 1.0);
  mat_specular : Array [0..3] of GLFloat = (1.0, 1.0, 1.0, 0.0);
  mat_shininess : GLFloat = 2.0;
begin
  glEnable(GL_DEPTH_TEST);
  glEnable(GL_AUTO_NORMAL);
  glEnable(GL_NORMALIZE);
  glEnable(GL_LIGHTING);
  glEnable(GL_LIGHT0);
  glEnable(GL_LIGHT1);
  glLightModeli(GL_LIGHT_MODEL_TWO_SIDE, 1);
  glLightfv(GL_LIGHT0, GL_AMBIENT, @ambient);
  glLightfv(GL_LIGHT0, GL_POSITION, @position1);
  glLightfv(GL_LIGHT1, GL_POSITION, @position2);
  glMaterialfv(GL_FRONT_AND_BACK, GL_DIFFUSE, @mat_diffuse);
  glMaterialfv(GL_FRONT_AND_BACK, GL_SPECULAR, @mat_specular);
  glMaterialfv(GL_FRONT_AND_BACK, GL_SHININESS, @mat_shininess);
  glEnable (GL_COLOR_MATERIAL);
  glClearColor (0.0, 0.75, 1.0, 1.0);
  glEnable(GL_MAP2_VERTEX_3);
  glMapGrid2f(20, 0.0, 1.0, 20, 0.0, 1.0);
  init_surface;
end;
procedure TfrmGL.Init_Surface; // собственно создание поверхности
var
 f : TextFile;   // текстовый файл хранящий координаты точек
 i : Integer;
 Model : TList;   // список модели
 wrkPatch : TPatch;  // переменная , представляющая собственно патч
 pwrkPatch : ^TPatch;  // указатель на патч
begin
 Model := TList.Create;    // создаем список модели
 AssignFile (f, 'Roza.txt');   // привязываем конкретный файл к переменной
 ReSet (f);       //  открываем для чтения
 While not eof (f) do begin      // если не конец
   For i := 0 to 24 do       // напомню – 25 патчей
      ReadLn (f, wrkPatch [i].x, wrkPatch [i].y, wrkPatch [i].z); // считываем значения
   New (pwrkPatch);   // создаем указатель
   pwrkPatch^ := wrkPatch;   // указываем указателю на весь массив сразу
   Model.Add (pwrkPatch); // все значения – в список модели ( для этого и нужен указатель)
 end;
 CloseFile (f);  // закрываем файл
 glNewList (ROZA, GL_COMPILE);  // а теперь создадим дисплейный список
   glPushMatrix;
   glScalef (0.5, 0.5, 0.5);
   For i := 0 to 11 do begin   // первые 12 патчей – лепестки
     glColor3f (1.0, 0.0, 0.0);  // они красные
     glMap2f(GL_MAP2_VERTEX_3, 0, 1, 3, 5, 0, 1, 15, 5, Model.Items[i]);  // расчет сетки
     glEvalMesh2(GL_FILL, 0, 20, 0, 20);  // выводим сетку
   end;
   For i := 12 to  Model.Count - 1 do begin  // остальные - стебель
     glColor3f (0.0, 1.0, 0.0); // он зеленый
     glMap2f(GL_MAP2_VERTEX_3, 0, 1, 3, 5, 0, 1, 15, 5, Model.Items[i]);
     glEvalMesh2(GL_FILL, 0, 20, 0, 20);
   end;
   glPopMatrix;
 glEndList;  // конец дисплейного списка
 Model.Free; // освободить список модели
end;
procedure TfrmGL.WMPaint(var Msg: TWMPaint); // процедура рисования
var
  ps : TPaintStruct;
begin    // здесь все должно быть по предыдущим примерам понятно
  BeginPaint (Handle, ps);
  glClear(GL_COLOR_BUFFER_BIT or GL_DEPTH_BUFFER_BIT);
  glPushMatrix;
    glRotatef(AngX, 1.0, 0.0, 0.0);
    glRotatef(AngY, 0.0, 1.0, 0.0);
    glRotatef(AngZ, 0.0, 0.0, 1.0);
    glCallList (ROZA);  // вызов листа
  glPopMatrix;
  SwapBuffers (DC);
  EndPaint (Handle, ps);
end;
procedure TfrmGL.FormResize(Sender: TObject);  // на случай изменение формы
begin  // зададим видовые параметры
 glViewport(0, 0, ClientWidth, ClientHeight);
 glMatrixMode(GL_PROJECTION);
 glLoadIdentity;
 gluPerspective (20.0, ClientWidth / ClientHeight, 1.0, 50.0);
 glMatrixMode(GL_MODELVIEW);
 glLoadIdentity;
 glTranslatef (0.0, 0.0, -18.0);
 InvalidateRect(Handle, nil, False);
end;
procedure TfrmGL.FormDestroy(Sender: TObject); // здесь все постаринке
begin
  glDeleteLists (ROZA, 1);
  wglMakeCurrent(0, 0);
  wglDeleteContext(hrc);
  ReleaseDC(Handle, DC);
  DeleteDC (DC);
end;
procedure TfrmGL.FormKeyDown(Sender: TObject; var Key: Word;Shift: TShiftState);
begin // нажатие клавиш
  Case Key of
    VK_ESCAPE : begin  Close; Exit;  end;
    VK_LEFT :   AngY := AngY + 5;
    VK_UP :     AngZ := AngZ + 5;
    VK_RIGHT :  AngX := AngX + 5;
  end;
  InvalidateRect(Handle, nil, False);
end;
procedure TfrmGL.FormCreate(Sender: TObject); создание формы
begin
  DC := GetDC(Handle);
  SetDCPixelFormat;
  hrc := wglCreateContext(DC);
  wglMakeCurrent(DC, hrc);
  Init;
end;
procedure TfrmGL.SetDCPixelFormat; // формат пиксела
var
  nPixelFormat: Integer;
  pfd: TPixelFormatDescriptor;
begin
  FillChar(pfd, SizeOf(pfd), 0);
  pfd.dwFlags := PFD_DRAW_TO_WINDOW or PFD_SUPPORT_OPENGL or
                 PFD_DOUBLEBUFFER;
  nPixelFormat := ChoosePixelFormat(DC, @pfd);
  SetPixelFormat(DC, nPixelFormat, @pfd);
end;

end.

